

こんにちは。私の名前は、Martyn Pike です。Ada University の大規模プログラミングの講義によ
うこそ。

この講義のテーマは、Adaのアクセス型です。最初に一連のスライドによる学習を終えた後、幾つかの
クイズを通して、学んだことを確認します。

スライドは、三つの部分に分かれています。最初は、メモリを割り当てるときの方法間の違いにつ
いて、二番目は、アクセス型の記述法の紹介、最後にアクセス型利用時の規則に従う方法について学
びます。



動的メモリ

Copyright © AdaCore

Slide: 2

アクセスされるオブジェクトは、実行時にどのようにメモリに割り当てられるかを理解することは、複雑な Ada のアクセス型を学ぶときに重要です。

アクセス型の設計

- Javaにおける参照や、C/C++ のポインタと同様のものを Ada ではアクセス（型）と呼びます
- オブジェクトは、メモリプールと関係します
- メモリプールが異なると、割り当て・解放のポリシーも異なります
- 検査なし解放を行わないことで、また、領域特有のアクセス型を用いることで、アクセスした値は、常に意味を持つことを保証できます
- Adaではアクセス型は、型です

```
type Integer_Access is access all Integer;  /* in C */
int * V = malloc (sizeof (int));
V : Integer_Access := new Integer;          /* or in C++ */
int * V = new int;
```

Copyright © AdaCore

Slide 3

Ada のアクセス型は、Java の参照や C/C++ におけるポインタと同じ基本的な特徴を持っています。

記憶域プール中のオブジェクトが参照できること、メモリの割り当て・解放を異なった方法できるように、これらのアクセス型は設計されています。

ポインタに共通する問題は、ポインタが示しているものが、常に正しく、かつ意味あるものであるということを保証できないことです。Ada では、この欠点を補うように、アクセス型を設計されています。

個々のアクセスが、異なったメモリ領域に及ばないことを保証するように、慎重に型付けしているというのが、Ada のアクセス型の主たる特徴です。

例にある左のコードでは、アクセス型宣言は、整数型オブジェクトのみを指し示すことができると宣言しています

次に、アクセス型のスタック領域オブジェクトの宣言です。ここでは、new キーワードを用いて、明示的な初期化として、オブジェクトをメモリ領域に割り当てています。

右側の、C 言語版では、malloc 関数を用いて、C++ 版 では new 演算子を用いて、同等の機能を実現しています。

アクセス型は危険である

- ・ メモリに関する複数の問題
 - リーク／メモリ不斉 (memory corruption)
- ・ 解析が難しい潜在的ランダム不全 (random failure) を招く
- ・ データ構造の複雑性の増大
- ・ アプリケーション性能の劣化の可能性
 - 間接参照は、直接参照よりもわずかにコストが増加する
 - スタック領域オブジェクトよりも、割り当てに、大きなコストがかかる
- ・ Ada言語では、可能な限りアドレスの利用を避けている
 - 配列はポインタではない
 - パラメータは、暗黙的に参照渡しである
- ・ 必要なときのみアクセス型を使用する

Copyright © AdaCore

Slide 4

Adaにおけるアクセス型のアプローチは、伝統的なポインタの問題を解決しますが、問題がなにもないというわけではありません。

様々な対処をしても、それでもやはり、プログラマがメモリリークやメモリ不斉を招くことがあります。

(ポインタにある) 動的な特性のために、欠陥の分析やデバッグが難しくなります。また、ランダムなふるまいのように見える場合があります。

アクセス型を通して、オブジェクトにアクセスする場合、2段階の手続きを経るため性能に影響がでる場合があります。このことは、直接アクセスよりわずかにコストが掛かることを示しています。


もちろん、実行時の動的なメモリ割り当ては、スタック領域上のオブジェクトを用いるよりも、より多くのコストが掛かります。これはメモリプールの管理とメモリの断片化といった問題に起因しています。

総じて Ada 言語ではメモリアドレスの使用を避けています。c 言語と異なり、配列はポインタではなく、パラメータは参照渡しとなります。


ポインタにまつわる先に挙げた危険性の幾つかは、とても深刻なものです。従って、いつもアクセス型を用いるのなら、必ず注意深く考えるようにする、どうしても必要だというとき以外は使用しないようにすべきです。

スタック領域とヒープ領域

```
I : Integer := 0;  
J : String := "Some Long String";
```



```
I : Access_Integer := new Integer'(0);  
J : Access_String := new String'("Some Long String");
```



一般的なポインタの落とし穴を完全に理解するために、実行時のメモリ割り当てに関する基本的な違いを、再度確認することは有益です。

伝統的に、2つのメカニズムがあります。最初はスタック領域で、二つ目はヒープ領域です。

Adaでは、宣言部分に現れる全てのオブジェクトのためにスタック領域を使用します。最初の例にあるI, Jは、スタック領域において連続する場所を占めています。

二番目の例では、アクセス型のオブジェクト自身は、スタック領域中にいますが、そのオブジェクトを通してアクセスする実際のオブジェクトは、ヒープ領域のどこかにいることを示しています。

少し手を休めて、この図のイメージを心に刻んで下さい。アクセス型について考える時、このイメージをどれだけ強調しても、強調しすぎることはありません。



Adaアクセス型

Copyright © AdaCore

Slide: 6

次に、Ada アクセス型の仕様に関する基礎的な設計原則と危険性のいくつかについて、見ていきます。

アクセス型全般

- 全ての記憶域プールをポイントすることができます（スタック領域を含む）

```
type T is [...]  
type T_Access is access all T;  
V : T_Access := new T;
```

- アクセス型はそれ自身個別の型です

```
type T_Access_2 is access all T;  
V2 : T_Access_2 := T_Access_2 (V);
```

- 注記：メモリプール特有のアクセス型（「all」予約語なし）があります。別の制約について、後ほど議論します

```
type T is [...]  
type T_Access is access T;  
V : T_Access := new T;
```

Copyright © AdaCore

Slide 7

さて、ここでは、アクセス型の構文を知り、その意味について考えます。

次の基本的な点から始めます。アクセス型は、個別の Ada の型であり、従って、名前・スコープ・属性を持ちます。

構文としては、「access all」キーワードを用い、次に、アクセスされるオブジェクトの型を示します。

例にあるコードの断片に示すように、T_Access や T_Access2 は T 型のオブジェクトにアクセスします、この両者は、Ada の型規則に適合する個別の型となります。

宣言から「all」キーワードを除いているメモリ領域特有のアクセス型宣言があります。「all」を含まない場合に関しては、のちほど説明します。

最後のコードの断片は、その例になります。

ヌル (null) 値

- 実際のデータをポイントしていないポインタは、ヌル値を持っているといいます
- 初期化しないと、デフォルトでポインタはヌルです
- 代入や比較で、ヌルを用いることができます

```
type Acc is access all Integer;  
  
V : Acc;  
begin  
  if V = null then  
    -- will go here  
  end if  
  
  V := new Integer'(0);  
  V := null; -- semantically correct, but introduces a leak
```

Copyright © AdaCore

Slide 6

C 言語や C++ 言語においてポインタを初期化するときに、数値ゼロを用いるか、NULL リテラル定数を用いるかは、いつも大きな論争になります。

Ada プログラミング言語は、ヌル (null) と呼ぶ予約語を持ち、言語の一部として、あいまいさがないように規定しています。

このことによって、コンパイラは、どのように使用されるかについての明確な前提を行うことができ、もし、ターゲットとなるオブジェクトが指定されていなければ、デフォルトのヌル値をアクセス型オブジェクトに対して与えます。

ヌルキーワードは、作成しようとする機能の明確な仕様を提供するために、比較と代入でも用いることができます。

この例題は、使った場合に危険となるヌルの利用について示しています。例の最後の行は、意味的には正しいのですが、不適切な場合を示しています。（前の行で）割り当てた整数オブジェクトの場所がもはや不明になり、解放することができません。古典的なメモリリークです。

メモリ割り当て

- ・「**new**」予約語を用いて、オブジェクトを作ります
- ・作られたオブジェクトは、割り当て時に加えられた制約を持ち続けます

```
V : String_Access := new String (1 .. 10);
```

- ・オブジェクトは、既存のオブジェクトをコピーすることによっても作ることができます。この時、修飾子を用います

```
V : String_Access := new String'("This is a String");
```

Copyright © AdaCore

Slide 9

これまでの説明で、「new」予約語を用いました。この語は、ターゲットとなるオブジェクトで、明示的にアクセス型を初期化するのに用います。

これまでのスライドを思い起こすと、new は、記憶領域プールのオブジェクトに対して割り当てを行いました。

「new」予約語に続いて、アクセス型のターゲットの型、その型が必要とする全ての制約を記述します。

割り当てによりメモリが実際に消費され、その消費量は事前に知る必要があるということは、道理にかなったことです。

修飾子は、ターゲット型となる既存のオブジェクトをコピーするために用います。デフォルトとなるオブジェクトが利用可能であると便利です。また、レコード型や配列型に対しては、集成代入 (aggregate assignment) を用いたいと思うでしょう。

解放 (Deallocation)

`with Ada.Unchecked_Deallocation;`

汎用副プログラムへの依存

`procedure P is
 type An_Access is access all A_Type;`

`procedure Free is new Ada.Unchecked_Deallocation (A_Type, An_Access);`

`V : An_Access := new A_Type;
begin
 Free (V);
end P;`

V は、呼び出し
後にヌルとなる

メモリ解放手続きの作成（汎
用副プログラムの実体化）

最初に型を、次にアクセス型
を記述する

割り当てたオブジェクトが不要になったときに、それらを記憶域プールに返す機構が、プログラム言語には必要です。

これは、メモリリークへ対処するための基本的な戦略です。Ada 言語は、賢く手際のよい手段を用いています。

いつ、どうやってオブジェクトが解放されるべきかを、開発者が完全に理解していることを、Ada 言語は要求しています。記憶域プールの各オブジェクトは全て、解放するために汎用体ライブラリを実体化することによって、（オブジェクトの解放を）行います。

開発者は、このことをやり過ぎと思うかもしれませんが、この労力は無駄になりません。また、アクセス型は、そうせざる終えないときのみ使用するということを思い出して下さい。

汎用手続きは、コールされた `Unchecked_Deallocation` であり、標準の Ada ライブラリから利用可能です。それは 2 つの汎用パラメータとともに、実体化されます。最初のパラメータは、ターゲットの型であり、二番目がアクセス型名になります。

このコード例には、`A_Type` というターゲット型とともに `An_Access` というアクセス型が宣言されています。次に、`Free` と云う名前で、`Unchecked_Deallocation` を実体化するのに、このアクセス型を（ターゲット型とともに）用いています。

アクセス型オブジェクトは、割り当てられたターゲットオブジェクトとともに明示的に初期化されます。（ここでの）唯一のボディ部の文は、`Free` の呼び出しであり、ターゲットオブジェクトを解放し、アクセス型オブジェクトを再度ヌルにセットしています。

ポインタの値参照 (Dereferencing)

- `.all` は、アクセスの逆参照を行います
 - ポインタによってポイントされるオブジェクトにアクセスします
- `.all` は、以下のものではオプションです
 - 配列の要素にアクセスする
 - レコードの要素にアクセスする

Copyright © AdaCore

Slide 11

一度、アクセス型オブジェクトにターゲットオブジェクトが与えられると、開発者は、アクセス型オブジェクトの値参照によって、ターゲットオブジェクトへのアクセスが可能になります。

「`.all`」予約語をアクセス型のオブジェクト識別子に加えることで、値参照を行います。これによってターゲットオブジェクトにアクセス出来るようになります。

これは、C 言語や C++ 言語におけるアスタリスクと同様のものです。

「`.all`」接尾辞は、ある環境下では、オプションです。即ち、配列の要素や、レコードの要素へアクセスする場合です。

値参照の例

```
type R is record
  F1, F2 : Integer;
end record;

type A_Int is access all Integer;
type A_String is access all String;
type A_R is access all R;

V_Int      : A_Int := new Integer;
V_String   : A_String := new String("abc");
V_R        : A_R := new R;

[...]

V_Int.all := 0;
V_String.all := "cde";
V_String(1) := 'z'; -- similar to V_String.all(1) := 'z';
V_R.all := (0, 0);
V_R.F1 := 1; -- similar to V_R.all.F1 := 1;
```

再帰構造を作るためにポインタを使用する

- ・ 再帰構造を宣言することはできません
- ・ しかし、包含している型へのアクセスは可能です

```
type Cell;
type Cell_Access is access all Cell;
type Cell is record
  Next      : Cell_Access;
  Some_Value : Integer;
end record;
```

部分的な宣言

完全な宣言

再帰データ構造は、プログラミングにおいて、大変に有益です。特に、連結リストやそれと同様のデータ構造において、役立ちます。

包含するスコープへのアクセス型オブジェクトを組み込むような賢い方法を使うことなしに、再帰データ構造を宣言することはできません。

スライドにある例は、伝統的な単一の連結リストです。最初に、Cell という型を部分的に宣言します。次に、その型へのアクセス型を宣言します。

次に、レコード型としての完全な宣言中に、一つのフィールドとして、Cell アクセス型の実体を組み込みます。

再帰的なデータ構造の最後（のデータ）を識別するために、比較における「null」予約語を用いると、容易になります。

スタック領域を参照する

- デフォルトでは、オブジェクトは参照できません。またコンパイラが、レジスタ中で最適化のためにオブジェクトを除く場合があります
- “**aliased**” により、アクセス型オブジェクトを通して参照するオブジェクトを宣言します
- “**'Access**” により、オブジェクトへの参照を得ることができます

```
V : aliased Integer;  
A : Int_Access := V'Access;
```

Copyright © AdaCore

Slide 14

これまで、動的に割り当てられたヒープオブジェクトにアクセスために使用するアクセス型について見てきました。

同様にスタック領域に割り当てられるアクセスオブジェクトに対しても、アクセス型を使用することができます。

最初のポイントは次です。スタック領域を用いて実体化する全てのオブジェクトには、何の前提もありません。実際、コンパイラはこれらのオブジェクトを配置するのに、自由にレジスタを用いることができます。

従って、アクセス型を用いて、これらオブジェクトにアクセスしようとする問題となります。

2つめのポイントは次です。オブジェクトを宣言するとき、オブジェクトに対して「**alias**」キーワードを付加します。そのオブジェクトを利用したい全てのアクセス型オブジェクトを、**Object'Access** という形式で初期化します。



アクセス可能性の検査

Copyright © AdaCore

Slide: 15

講義の最後となるこのセクションでは、アクセス可能性に関する規則について説明します。このセクションが終わったあと、確認のクイズを始めます。

アクセス可能性の検査に関する課題

- ・ 次の状況について考えます

```
type Acc is access all Integer;  
G : Acc;  
  
procedure P is  
  V : aliased Integer;  
begin  
  G := V'Access;  
end P;
```

- ・ 次は安全でしょうか

```
G.all := 0;
```

- ・ Adaでは、上記の文は禁止されています

アクセス可能性レベル規則は、何らかの理由で不正となっているターゲットオブジェクトにアクセスしないように設計されています。この規則を破ることがないように、十分に注意しなくてはなりません。

ここでは、整数に対するアクセス型を宣言し、Gとして、アクセス型オブジェクトの実体を宣言している場合について考えます。

手続きPは、別名化した整数Vスタック領域のオブジェクトを宣言し、Gがそのオブジェクトにアクセスできるように宣言しています。

ここでは、手続きPの外側で、G.allを用いています。これは安全でしょうか。

Adaでは、アクセス可能性レベルに関する規則を破っているため、このコードを禁止しています。

Gの有効期間は、Vより長いため、手続きPが終了したとき、スタック領域のオブジェクトであるVは、もはや正当ではありません。副プログラムが終了したときには、関係するスタック領域は、もはや存在しないからです。

それゆで、Gの値参照は、もはやスコープ中には存在しないターゲットオブジェクトにアクセスしようとしていることになります。

ネストレベルの定義

- オブジェクトのアクセス可能性レベルは、副プログラム階層における深さと同じです

```
package body P is
  V : aliased Integer; -- Level 0 / Library-Level

  procedure Proc is
    V : aliased Integer; -- Level 1

    procedure Nested is
      V : aliased Integer; -- Level 2
    begin
      null;
    end Nested;
  begin
    null;
  end Nested;
end P;
```

- ライブラリレベルで宣言されるアクセス型は、**level 0** のオブジェクトのみをポイントすることができます

理解すべき重要な概念に、ネストレベルがあります。ネストレベルは、プログラムのソースコードの構造化に用いる入れ子と同様にスコープのレイヤによって定まります。

Ada では、アクセス可能性レベル (accessibility level) という用語を用います。入れ子において、もっとも外側のスコープであるレベル 0 と比較して、あるコードを包含しているスコープのレベルを示す用語です。

このコードの例では、レベル 0 からレベル 2 までの 3 つのアクセス可能性レベルがあります。

規則は次です。ライブラリレベルのコードは、アクセラレベル 0 のオブジェクトにアクセスできます。他のレベルにいるコードは、アクセスオブジェクト自身のスコープ中に存在し続けるアクセスターゲットオブジェクトにアクセスできます。通常、このことは、(アクセスオブジェクト) 自身を包含しているスコープの全てを意味します。

各包含されているスコープでは、最後に、スタックは解放されるため、この規則には意味があります。包含している (側の) スコープ中にいるアクセスオブジェクトは、解放されたスタック領域中のオブジェクトをターゲットとすることになり、不正なアクセスとなるからです。

アクセス型とアクセス可能性検査

```
package body P is
  type T0 is access all Integer;
  A0 : T0;
  V0 : aliased Integer;

  procedure Proc is
    type T1 is access all Integer;
    A1 : T1;
    V1 : aliased Integer;
  begin
    A0 := V0'Access;
    A0 := V1'Access;
    A0 := V1'Unchecked_Access;

    A1 := V0'Access;
    A1 := V1'Access;
    A1 := T1 (A0);
    A0 := T0 (A1);

    A1 := new Integer;
    A0 := T0 (A1);
  end Proc;
end P;
```

- こういった問題に直面しないよう、ネスト化したアクセス型は避けること

Copyright © AdaCore

Slide 18

ネスト化したアクセス型を極力避けるというのが、これまでのスライドで学ぶべき重要な点です。

例では、Level 0 において、整数ターゲットオブジェクトに対するアクセス型の宣言があります。

A0 として実体を作り、次に、別名化した整数スタック領域オブジェクトを宣言しています。

手続き Proc では、整数ターゲットオブジェクトに対するアクセス型を宣言しています。次に、この実体と別の別名化したスタック領域オブジェクトを宣言しています。

アクセス型オブジェクトは、「' Access」属性を用いることによって、別名化したスタック領域オブジェクトにアクセスできることを思い出して下さい。

ターゲットオブジェクト V1 に対して、より高いレベルの A0 が宣言されているということが、最初の誤りです。

二番目のエラーも同様です。アクセス型に型変換していますが、ターゲットオブジェクトは、アクセス型オブジェクトよりも低いレベルで宣言されています。

三番目のエラーも同様で、ターゲットオブジェクト A1 が、アクセス型オブジェクト A0 よりもより低いレベルであるためです。

アクセス可能性検査を迂回する

- 安全ではないデータへのアクセスも、時には問題ないことがあります
- 「'Unchecked_Access」属性を用いると、適合しないアクセス可能性レベルの変数にアクセスすることが可能です
- 潜在的な問題には、注意して下さい

```
type Acc is access all Integer;  
G : Acc;  
  
procedure P is  
  V : aliased Integer;  
begin  
  G := V'Unchecked_Access;  
end P;
```

Copyright © AdaCore

Slide 19


先のスライドの例で、「'Unchecked_Access」と呼ばれる 'Access 属性の変種を用いていることに気がついたかもしれません。

データへの安全ではないアクセスを行ったときに、開発者の責任で、いつなら、あるいはどうして問題がないかを判断しなくてはなりません。

Ada は、「'Unchecked_Access」属性によって、アクセス可能性レベル規則を意図的に破ることを認めています。また、アクセス可能性レベルが不適合であるスタック領域のオブジェクトにアクセスすることも認めています。

この属性を利用することは危険です。危険性を上手に伝えることはできませんが、このメッセージを受け止めてくれることを望んでいます。

もし、プログラムソースコードを読んだときに「'Unchecked_Access」を利用する場所に出会ったら、よりよい方法を知らないプログラマによって利用されているのではないかと、注意するようにしてください。



いよいよこの講義の最後になりました。

既に、Ada 言語のアクセス型について十分な知識を得たと思います。理解を試すためのちょっとしたクイズをやしましょう。

幸運を！

?

正しいですか (1/10)

✓

はい
(チェックアイコンをクリックする)

いいえ
(エラーの場所をクリックする)

```
type An_Access is access all Integer;  
  
W : Integer;  
V : An_Access := W'Access;
```

Copyright © AdaCoreSlide 21

ゆっくりとはじめます。次のコードが正しいかどうかを判断して下さい。

アクセス型は、整数ターゲット型に対して宣言されています。

整数 W を宣言し、'Access属性を、明示的に An_Access 型のオブジェクトである V を初期化するために用いています。

このコードは正しいでしょうか。

もし、コードが正しいと思えば、「はい」のアイコンをクリックして下さい。或いは、間違っていると思うコードの場所をクリックして下さい。

? 正しいですか (1/10) ✖ いいえ

```
type An_Access is access all Integer;  
W : Integer;  
✖ V : An_Access := W'Access;
```

アクセス型からアクセスするためには、Wを別名化する必要があります

このコードは間違っています。

'Access 属性の利用は、「aliased」予約語を用いて宣言されたスタック領域オブジェクトに制限されています。

この例において、Wは、「aliased」予約語を用いて宣言されていないので、アクセス型からアクセスすることができません。

?

正しいですか (2/10)

✓

はい
(チェックアイコンをクリックする)

いいえ
(エラーの場所をクリックする)

```
type An_Access is access Integer;  
W : aliased Integer;  
V : An_Access := W'Access;
```

Copyright © AdaCoreSlide 23

この2番目の問題では、先のコードを少し変更しました。もう一度同じ質問をします。

このコードは正しいでしょうか。

もし、コードが正しいと思えば、「はい」のアイコンをクリックして下さい。或いは、間違っていると思うコードの場所をクリックして下さい。

? 正しいですか (2/10) ✖ いいえ

✖ `type An_Access is access Integer;`
`W : aliased Integer;`
`V : An_Access := W'Access;`

アクセス型は、別名化した変数をポイントするために「all」が必要です。
今後は、常に「access all」を用いることを推奨します

このコードは、間違っています。

アクセス型オブジェクトが、スタック領域オブジェクトをターゲットとするためには、アクセス型を、「all」予約語と共に宣言しなければなりません。

柔軟性を増すために、開発者は、つねに アクセス型宣言において、access all メソッドを使用することを勧めます。



正しいですか

(3/10)



はい
(チェックアイコンをクリックする)

いいえ
(エラーの場所をクリックする)

```
type An_Access is access all Integer;  
  
procedure Proc is  
  W : aliased Integer;  
  X : An_Access := W'Access;  
begin  
  null;  
end Proc;
```

この問題でも、整数ターゲット型に対する An_Access アクセス型を、引き続き利用します。

宣言部には、二つのスタック領域オブジェクトがあり、ボディ部には空の手続き Proc があります。

W は、別名化した整数型であり、X は、W の ' Access 属性を利用し、明示的に初期化したアクセス型の実体です。

もし、コードが正しいと思えば、「はい」のアイコンをクリックして下さい。或いは、間違っていると思うコードの場所をクリックして下さい。



正しいですか

(3/10)



いいえ



```
type An_Access is access all Integer;  
  
procedure Proc is  
  W : aliased Integer;  
  X : An_Access := W'Access;  
begin  
  null;  
end Proc;
```

X はアクセス可能性レベルがゼロです。W はアクセス可能性レベルが 1 です。
可能な宣言は、現在のスコープの外にあるオブジェクトへの参照です

残念なことにこのコードは正しくありません。アクセス可能レベル規則を破っているため、コンパイルすることができません。

x のアクセス可能性レベルは 0 です。その型が宣言されているレベルからそうなります（一行目）。一方で、w のアクセス可能性レベルは 1 です。

この宣言で可能になるのは、現在のスコープの外側にあるオブジェクトへの参照になります。



正しいですか (4/10)



はい
(チェックアイコンをクリックする)

いいえ
(エラーの場所をクリックする)

```
type R is record
  F1, F2 : Integer;
end record;

type R_Access is access all R;

procedure Proc is
  V : R_Access := new R;
begin
  V.F1 := 0;
  V.all.F2 := 0;
end Proc;
```

ここでは、レコード型 R と関連するアクセス型を見ます。

手続き Proc は、動的に割り当てられた R の実体にアクセスするために、明示的に初期化されたアクセス型オブジェクトを宣言しています。

手続き Proc のボディ部で、R の 2 つの要素のうち、一つは「.all」予約語を用いて、他方は用いていません。

もし、コードが正しいと思えば、「はい」のアイコンをクリックして下さい。或いは、間違っていると思うコードの場所をクリックして下さい。



正しいですか (4/10)



はい

```
type R is record
  F1, F2 : Integer;
end record;

type R_Access is access all R;

procedure Proc is
  V : R_Access := new R;
begin
  V.F1 := 0;
  V.all.F2 := 0;
end Proc;
```

V の要素に直接、或いは .all を用いてアクセスすることができます

しかし、割り当てられた R はメモリーリークします。

このコードは構文的には正解です。アクセス型オブジェクトは、.all 接尾辞のあるなしに関わらずターゲットオブジェクトの要素を参照することが可能です。

しかし、このコードには明らかなメモリーリークがあります。動的に割り当てられた R の実体は、記憶域プールに返ることができないからです。Proc 手続きが終了するとき、V スタック領域オブジェクトは失われます。その場合、ソースコードの他の部分は、V がターゲットとしていたオブジェクトを解放することができません。

?

正しいですか

(5/10)

✓

はい
(チェックアイコンをクリックする)

いいえ
(エラーの場所をクリックする)

```
G : aliased Integer;  
  
procedure Proc is  
  type A_Access is access all Integer;  
  V : A_Access;  
begin  
  V := G'Access;  
end Proc;
```

Copyright © AdaCore

Slide: 26

5 番目の問題は、G と呼ぶ、レベル 0 の整数別名化スタック領域オブジェクトの宣言に関するものです。

手続き Proc は、レベル 1 であり、スタック領域を含む全ての記憶域から指し示すことのできる整数オブジェクトへのアクセス型を宣言しています。

このアクセス型の実体は、オブジェクト V として宣言されています。唯一のボディ部の文は、レベル 0 のスタック領域オブジェクト G に対して、' Access 属性を用いて、V のターゲットとしています。

このコードは正しいでしょうか。

もし、コードが正しいと思えば、「はい」のアイコンをクリックして下さい。或いは、間違っていると思うコードの場所をクリックして下さい。



正しいですか

(5/10)



はい

```
G : aliased Integer;  
  
procedure Proc is  
  type A_Access is access all Integer;  
  V : A_Access;  
begin  
  V := G'Access;  
end Proc;
```

A_Access はレベル 1 です。G のようなレベル 0 のオブジェクトをポイントすることができます。A_Access のスコープは狭いのです。

これは正しいコードです。アクセス可能性規則により、包含しているスコープ中で宣言されたスタック領域オブジェクトへのアクセスが可能です。

アクセス型 A_Access は、レベル 1 に居ます。これは、G のような、スコープに関して低いレベルであるレベル 0 として宣言されたオブジェクトを、ターゲットとすることができます。

規則によれば、包含しているスコープ中で宣言されたオブジェクトは、包含されているスコープからアクセス可能ということを思い出して下さい。



正しいですか (6/10)



はい
(チェックアイコンをクリックする)

いいえ
(エラーの場所をクリックする)

```
type R is record
  F1, F2, F3 : Integer;
end record;

type R_Access is access all R;
type R_Access_Access is access all R_Access;

V : R_Access_Access;
begin
  V := new R_Access;
  V.all := new R;
  V.F1 := 0;
  V.all.F2 := 0;
  V.all.all.F3 := 0;
```

次の問題は、アクセス型をターゲット型としている別のアクセス型、という少し複雑なケースについて考えます。

ここには、R という単純なレコード型宣言があり、ターゲット型が R であるアクセス型宣言が続きます。

次に、この宣言されたアクセス型 R_Access をターゲット型とするアクセス型を宣言します。

更に、R アクセス型を二重にターゲットとするスタック領域オブジェクトを作ります。

ボディ部には、V にアクセスしたり、操作したりする様々な文があります。

もし、コードが正しいと思えば、「はい」のアイコンをクリックして下さい。或いは、間違っていると思うコードの場所をクリックして下さい。



正しいですか

(6/10)



いいえ

```
type R is record
  F1, F2, F3 : Integer;
end record;

type R_Access is access all R;
type R_Access_Access is access all R_Access;

V : R_Access_Access;
begin
  V := new R_Access;
  V.all := new R;
  V.F1 := 0;
  V.all.F2 := 0;
  V.all.all.F3 := 0;
```

V.F1 は機能しません。一つの間接参照のみを省略できます。しかし、V はポインタのポインタです

ほとんどの文は合っています。しかし、コンパイルエラーを生じる文が一つあります。

V.F1 で始まる行は、不正です。は二重にターゲット化したアクセス型において、「.all」を省略することはできないからです。

「.all」は、ターゲットオブジェクトの要素にアクセスするときのみ省略できることを思い出して下さい。この場合、ターゲットオブジェクトは、それ自身アクセス型であり、何ら要素を持っていません。

?

正しいですか (7/10)

✓

はい
(チェックアイコンをクリックする)
いいえ
(エラーの場所をクリックする)

```
type A_Access is access all Integer;  
  
procedure Free is new Ada.Unchecked_Deallocation  
(Integer, A_Access);  
  
V1 : A_Access := new Integer;  
V2 : A_Access := V1;  
begin  
  Free (V1);  
  Free (V2);  
end;
```

Copyright © AdaCoreSlide 33

7 番目の問題は、割り当てられたオブジェクトを記憶域プールに戻すことに関する知識を問う問題です。

このコードは、アクセス型を宣言することから始まっています。このアクセス型は、スタック領域を含む全ての記憶域プールの全ての整数オブジェクトをターゲットにしています。

次に、汎用手続き `Unchecked_Deallocation` にターゲット型とアクセス型を引数として与え、`Free` として、宣言しています。

アクセス型の二つのスタック領域オブジェクトがあります。最初は、明示的に整数型で割り当て初期化を行った形で宣言されています。二番目は、最初のターゲットオブジェクトを使って、明示的に初期化を行った形で宣言されています。

ボディ部には、`V1` を引数にした `Free` 呼び出しと、`V2` を引数にした `Free` 呼び出しがあります。

このコードは正しいでしょうか。

もし、コードが正しいと思えば、「はい」のアイコンをクリックして下さい。或いは、間違っていると思うコードの場所をクリックして下さい。



正しいですか (7/10)



いいえ

```
type A_Access is access all Integer;

procedure Free is new Ada.Unchecked_Deallocation
(Integer, A_Access);

V1 : A_Access := new Integer;
V2 : A_Access := V1;
begin
  Free (V1);
  Free (V2);
```

V1 と V2 は同じオブジェクトを指しています。二番目の Free は、
(そうすることを期待しますが) `Storage_Error` を送出すべきです。

このコードをコンパイルできます。しかし、実行時に、二番目の Free 呼び出しは、`STORAGE_ERROR` 例外を送出します。

V1 と V2 は、両方とも同じオブジェクトを指しています。最初の Free が完了したとき、ターゲットオブジェクトは、記憶域プールに戻ります。しかし、V2 は、依然として（記憶域プールに戻ってしまった）オブジェクトを指したままです。



正しいですか (8/10)



はい
(チェックアイコンをクリックする)

いいえ
(エラーの場所をクリックする)

```
type A_Access is access all Integer;  
  
procedure Free is new Ada.Unchecked_Deallocation  
(Integer, A_Access);  
  
V : A_Access;  
begin  
  Free (V);  
  V := new Integer;  
  Free (V);  
  Free (V);
```

前回の問題にあったコードをもとに、別のコードを作ります。ボディ部の文を変えています。Free の呼び出しの後、確保した整数を V へ代入する文が続きます。

次に、2 つの連続する Free 呼び出しが続きます。

もし、このコードが正しいと思えば、「はい」のアイコンをクリックして下さい。或いは、間違っていると思うコードの場所をクリックして下さい。



正しいですか

(8/10)



はい

```
type A_Access is access all Integer;  
  
procedure Free is new Ada.Unchecked_Deallocation  
(Integer, A_Access);  
  
V : A_Access;  
begin  
  Free (V);  
  V := new Integer;  
  Free (V);  
  Free (V);
```

V は、null に初期化されています。従って、最初の free では何も起きません。2番目は実際にオブジェクトの解放を行い、null を設定します。最後は、再度 null を引数に呼び出しています

Copyright © AdaCore

Slide 36

これは正しいコードです。

明示的に初期化されていないVを引数にして、Freeを呼び出すことは可能です。全てのアクセス型オブジェクトは、デフォルトでは、nullに初期化されます。

Vが、いったんターゲットオブジェクトを割り当てられている場合、Freeに渡されると、ターゲットオブジェクトを記憶域プールに戻します。

Freeは、nullをアクセス型オブジェクトにセットします。従って、次の連続する2つのFree呼び出しは、規則に適合し、正当です。

記憶すべき重要なことは、nullとともに呼び出すFreeには、意味がないということです。



正しいですか

(9/10)



はい
(チェックアイコンをクリックする)

いいえ
(エラーの場所をクリックする)

```
type A_Access is access all Integer;  
  
procedure Free is new Ada.Unchecked_Deallocation  
(Integer, A_Access);  
  
V : A_Access;  
W : aliased Integer;  
begin  
  V := W'Access;  
  Free (V);
```

最後から2番目の問題は、再度 A_Access アクセス型を取り上げます。このアクセス型は、スタック領域を含む全ての記憶域プールの整数オブジェクトをターゲットにします。

Unchecked_Deallocation を、アクセス型とターゲット型の Free 手続きとして実体化します。

次にアクセス型 V の実体と、別名化整数 W の実体を宣言しています。

ボディ部で、Vは、Wへのアクセスを割り当てています。次に、Free にこの V を引数として渡します。

もし、コードが正しいと思えば、「はい」のアイコンをクリックして下さい。或いは、間違っていると思うコードの場所をクリックして下さい。



正しいですか

(9/10)



いいえ

```
type A_Access is access all Integer;  
  
procedure Free is new Ada.Unchecked_Deallocation  
(Integer, A_Access);  
  
V : A_Access;  
W : aliased Integer;  
begin  
  V := W'Access;  
  Free (V);
```

ここでは、スタック領域中のオブジェクトを解放しようとしています。それはできません

このコードは不正です。

スタック記憶域プール中のターゲットオブジェクトを解放することは不正です。この場合、w はスタック領域オブジェクトなので、v はスタック記憶域プール中のオブジェクトを指していることになります。

?

正しいですか (10/10)

はい
(チェックアイコンをクリックする)

いいえ
(エラーの場所をクリックする)

```
type A_Access is access all Integer;  
  
type R is record  
  V : A_Access;  
  W : aliased Integer;  
end record;  
  
G : R;  
  
procedure P is  
  L : R;  
begin  
  G.V := G.W'Access;  
  L.V := L.W'Access;  
end P;
```

Copyright © AdaCoreSlide 36

最後の問題では、スタック領域を含む全ての記憶域プール中の整数ターゲットオブジェクトに対するアクセス型を宣言しています。

次に、アクセス型の実体と別名化整数の実体を要素とするレコード型を宣言しています。

レコード型の実体として、オブジェクト G があります。

手続き P は、その内部に L と呼ぶ、別のレコード型実体を持っています。

P のボディ部では、二つの文があります。ここでは、アクセス型の要素を別名化整数要素へのターゲットオブジェクトとしています。

もし、コードが正しいと思えば、「はい」のアイコンをクリックして下さい。或いは、間違っていると思うコードの場所をクリックして下さい。

? 正しいですか (10/10) ✖

いいえ

```
type An_Access is access all Integer;  
  
type R is record  
  V : An_Access;  
  W : aliased Integer;  
end record;  
  
G : R;  
  
procedure P is  
  L : R;  
begin  
    G.V := G.W'Access;  
    L.V := L.W'Access;  
end P;
```

✖

L は、局所オブジェクトです。そのアクセス可能性レベルは 1 です
V は、レベル 0 のアクセス可能性レベルを持つアクセス型です


Copyright © AdaCore

Slide 40

このコードは正しくありません。

2 番目のボディ部の文は、アクセス可能性規則を破っています。入れ子の中で、低い位置から、アクセス型宣言を用いようとしています。

V と名付けられた要素は、レベル 0 のアクセス可能レベルであるアクセス型です。



Ada 大規模プログラムのアクセス型のコースに参加いただきありがとうございました。

Ada プログラム言語を学ぶ上で、この講義が貴重な一ステップとなったこと、Ada University における他のコースも引き続き受けられることを期待しています。

ありがとうございました。